
1 

PharmaSUG China 
 

Monitoring Child Growth and Safety Profile using Growth Charts 

Rajesh Babu Moorakonda, Singapore Clinical Research Institute, Singapore 

Mihir Gandhi, Singapore Clinical Research Institute, Singapore 

 
 

ABSTRACT  

In child nutritional trials, graphical presentation of subject-level anthropometric parameters over time is useful to 
visualize child growth trajectory according to age. The trajectories of selected anthropometric parameters are usually 
evaluated with reference to a growth chart developed using growth standards, such as World Health Organization 
(WHO) child growth standards. Growth charts provide expected percentile values of the anthropometric parameters 
based on children of same gender and age-group from the general population. It is usually important in child 
nutritional trials to monitor the child growth for potential effect of study product feeding, adverse events and 
concomitant medications. This paper describes step-by-step procedure for plotting subject-level trajectory of an 
anthropometric parameter during the trial period on the growth chart developed using WHO child growth standards. 
The chart also displays details of onset of adverse events as well as intake of concomitant medications. It uses 
PROC GPLOT and SAS Annotate facilities. 

INTRODUCTION  

Human growth can be divided into the phases of fetal, infant, childhood and pubertal growth. These phases overlap in 
time and interact with each other, and consequently the growth and development during one phase may influence the 
other. This potentially makes monitoring of child growth during infancy clinically important. For example, there is 
evidence that rapid rate of weight gain during infancy may lead to obesity later in the childhood and could result in 
cardiovascular diseases later in the life [1].  

A typical child growth chart consists of smooth curves of an anthropometric parameter (e.g., weight, height, upper 
arm circumference) plotted against age or another anthropometric parameter. The curves represent a set of 
percentile values (e.g., 97

th
, 90

th
, 75

th
, 50

th
, 25

th
, 10

th
 and 3

rd
 percentiles) or values corresponding to standard 

deviation scores (e.g., 1, 2, 3 SD above/below reference mean or median value) of the parameter based on data of 
same gender at various ages based on local or multi-country general population. 

For child nutritional trials, it is usually important to monitor child growth for various anthropometric parameters with 
reference to child growth charts developed using child growth standards based on local or close to local population. 
When local population based child growth standards are not available, often the World Health Organization (WHO) 
child growth standards, developed based on multi-country data, is preferred as the reference [1]. To monitor 
individual study subject’s growth in context of an anthropometric parameter, measurement values of the parameter 
are plotted on the corresponding child growth chart of that parameter (e.g., using WHO child growth standards) and 
compared with its 50

th
 percentile curve across various time points (age at the measurement collection time points). If 

the subject’s own measurement values systematically fall above the 50
th
 percentile of the growth chart, it is inferred 

that the subject had gained weight more than expected according to the reference population and vice versa.  
However, it may not be considered as clinically significant weight gain (or loss) unless the subject’s own weight 
crosses the pre-specified higher (or lower) percentile of the growth chart (e.g., 75

th
 percentile) or pre-specified margin 

of standard deviation score (e.g., 2 SD) above (or below) the 50
th
 percentile value. Usual expectation is that subjects 

should stay relatively consistent in their growth pattern. In nutritional trials, adverse events and/or intake of 
concomitant medications could also affect child growth. Hence, often clinicians are interested to see onset of adverse 
events (AEs) and intake of concomitant medication on the subject-level trajectory of anthropometric parameters 
during the trial period. In this paper, our aim is to demonstrate a step-by-step procedure for plotting subject-level 
trajectory of an anthropometric parameter on a growth chart developed using the WHO child growth standards and 
also displaying onset of AEs and intake of concomitant medications with their descriptions on the same chart. The 
final chart would look like the graph below:  

The bottom most dotted curve touching at -3 SD on the secondary Y-axis shows reference values for weight-for-age 
according to WHO child growth standards corresponding to -3 SD below 50

th
 percentile value. Similarly, the other 

dotted lines show expected weight-for-age at various standard deviation scores higher/below 50
th
 percentile value for 

the same gender children in the general population. The black solid line is the subject’s own weight trajectory over the 
trial period. Texts in red color show adverse event description and adjacent red vertical dotted lines show 
corresponding adverse event onset time-points. Similarly, the texts and vertical dotted lines in blue color show 
information on intake of concomitant medications.  

KEY WORDS: Growth charts, Safety profile, PROC GPLOT, Axis statement, Symbol statement, SAS annotation. 

DATA OVERVIEW 

In this paper, we are considering body weight as the anthropometric parameter of interest for male children (SEX=1) 
during birth to 180 days of age.  

1. For plotting WHO child growth standards reference lines (median and ±3 SD, ± 2 SD, ± 1 SD from median), 
we need dataset ‘weianthro.sas7bdat‘ having variables SEX, AGE, L, M, S (explained later in detail). This 
can be downloaded from WHO Child Growth Standards - SAS igrowup package from the WHO website [2]. 

2. To get the subjects safety profile as well as body weight variable in to one dataset, we require: 

Dataset A: Subject level dataset with unique subject identifier (USUBJID), gender (SEX), date of birth 
(BRTHDT) and other study product related information - study product start date (TRTSTDT), study product 
stop date (TRTENDT), early study termination date (EARLDT); 

 

Dataset B: Adverse events dataset with USUBJID, AE onset date (AESTDT), AE preferred term (AEPT), a 
flag variable (AESER) to indicate an AE as a serious adverse event (SAE) where (AESER=’YES’ for SAEs); 

Figure 1: Subject’s weight over age with reference to the WHO child growth standards  

Dataset C: Concomitant medications dataset with USUBJID, concomitant medication start date (CMSTDT); 
concomitant medication generic name (CMTERM); 

 

Dataset D: Anthropometric dataset with USUBJID, subject weight in kg (WT) and weight measurement date 
(WTDT) for each weight data collection study visits. 

 

In the each of the above datasets (A, B, C, D), add a variable AGE using formula AGE=VARDT – BRTHDT. 
Here VARDT represents TRTSTDT, TRTENDT, and EARLDT for dataset A; AESTDT for dataset B; 
CMSTDT for dataset C and WTDT for dataset D. In order to add AGE variable corresponding to variables 
TRTSTDT, TRTENDT, EARLDT one dataset, one may need to first create three datasets, say A1, A2 and 
A3 with AGE variable corresponding to TRTSTDT, TRTENDT, EARLDT and then merge in to single dataset 
using AGE as ‘by’ variable. During this process, do create 3 flag variables (SPSTART, SPSTOP, and 
EARLTERM) with the value ‘YES’ to indicate value in AGE variables either correspond to TRTSTDT, 
TRTENDT or EARLDT. Merge the updated dataset A, B, C and D into a single dataset, say ALL, using AGE 
as ‘by’ variable. The dataset ALL would have the structure displayed below: 

                 
 

PROGRAMMING OVERVIEW 

CONSTRUCTION OF WHO GROWTH STANDARDS REFERENCE LINES 

WHO growth standards are developed to provide standard deviation scores (also called Z-scores) using on LMS 
method [3]. WHO growth standards provide values of L, M and S parameters for series of age-groups for both male 
and female children. Thus, by supplying subject’s weight, age and gender to LMS method formula, Z-score = 
[(Weight/M)

L-1
]/(L*S), one can obtain Z-score. It is useful to assess whether the subject weight is higher or lower than 

the median weight of children from general population with same gender and age-group, and to what extent. 
Alternatively, we can also calculate weight corresponding to a given Z-score using formula Weight = M(1 + L*S*Z-
score)

1/L
 and  compare with subject’s weight. For comparing subject’s weight at various study visits, we use the later 

approach. We calculate the weights corresponding to 0, ±1, ±2 and ±3 WHO child growth standards Z-score for male 
population at ages corresponding time from birth to the study visits. These serve as WHO child growth standards 
reference lines. The line corresponding to 0 Z-score represents 50

th
 percentile (median) of population weight for male 

children.   

Below are sample data steps to calculate weights for plotting the reference lines:   

libname _reflib "C:\WHO\igrowup_sas"; 

data sdz(keep=SEX AGE sd:); 

  set _reflib.weianthro(where=(SEX=1 and AGE le 180)); 

  array sdx(7) sd3neg sd2neg sd1neg sd0 sd1pos sd2pos sd3pos; 

  do _i=-3 to 3; 

    sdx[_i+4]=m*((1+_i*l*s)**(1/l)); 

  end; 

run; 

 

Now, the reference lines are plotted using PROC GPLOT statements mentioned as below: 

proc gplot data=sdz gout=myplot; 

   plot (sd3neg sd2neg sd1neg sd0 sd1pos sd2pos sd3pos)*AGE/overlay frame 

name="sdplt"; 

run; 

quit; 

 

Below is a sample output: 

 

 

 

The display settings of the plotted reference lines can be changed using SYMBOL statements mentioned below:  

symbol1 interpol=spline width=1 color=brown  line=33; 

symbol2 interpol=spline width=1 color=red    line=33; 

symbol3 interpol=spline width=1 color=orange line=33; 

symbol4 interpol=spline width=1 color=green  line=33; 

symbol5 interpol=spline width=1 color=orange line=33; 

symbol6 interpol=spline width=1 color=red    line=33; 

symbol7 interpol=spline width=1 color=brown  line=33; 

 
Here SYMBOL1 is used for sd3neg, SYMBOL2 is for sd2neg and so on as mentioned in the PROC GPLOT step 
above. 

The X and Y axis display characteristics can be adjusted by adding AXIS statements mentioned below: 

Figure 2: Using GPLOT procedure with OVERLAY option, we can plot multiple variables against 

the single time variable (AGE) in same graph 

axis1 order=(0 to 180 by 10) major=(height=.3) length=87 pct minor=none 

offset=(1,0) label=(height=1 'Age (days)'); 

axis2 order=(0 to 14 by 1) minor=(height=.2 number=1) offset=(0,0) label=( angle=90 

height=1 'Weight (kg)'); 

 
Now the PLOT statement should be updated in the PROC GPLOT statements above to reflect the changes in axis 
settings. It should be now:  

plot (sd3neg sd2neg sd1neg sd0 sd1pos sd2pos sd3pos)*AGE / overlay frame 

name="sdplt" haxis=axis1 vaxis=axis2; 

 

Below is the sample output after updating PLOT statement.  

 

 
 
 

Now, we name the reference lines with the same color using the annotation facility. 

/* Create the Annotate data set LINETEXT */ 

data linetext;    

   length function color $ 10 text $ 30 style $ 25; 

   retain xsys '2' ysys '1' hsys '1' position '6'  size 1; 

          

   set all1 end=last; 

   style = "'Albany AMT/bold'"; 

   cborder=''; 

 

   if last then do; 

      xsys='2'; ysys='2'; hsys='3'; position='6'; size=2.5; 

      x=age;angle=50; 

      y=sd3neg; text=' -3 SD';  color='brown';  output; 

      y=sd2neg; text=' -2 SD';  color='red';    output; 

      y=sd1neg; text=' -1 SD';  color='orange'; output; 

      y=sd0;    text=' Median'; color='green';  output; 

Figure 3: Modified the display settings by SYMBOL and AXIS statements 

      y=sd1pos; text=' +1 SD';  color='orange'; output; 

      y=sd2pos; text=' +2 SD';  color='red';    output; 

      y=sd3pos; text=' +3 SD';  color='brown';  output; 

   end; 

run; 

 

Again, the PLOT statement in the PROC GPLOT step should be updated to reflect the above changes: 

plot (sd3neg sd2neg sd1neg sd0 sd1pos sd2pos sd3pos)*AGE / overlay frame name="sdz" 

      haxis=axis1 vaxis=axis2 annotate=linetext; 

 

Below is the sample output after updating PLOT statement.  

 

 

 

 

 

ADDING SUBJECT WEIGHT AND SAFETY PROFILE 

Now it’s time to add subject information on the plot. For this, the standard deviations dataset (SDZ) should be merged 
with subject profile dataset (ALL) by AGE. It also requires adding the subject weight variable (WT) to the PLOT 
statement. Moreover, additional SYMBOL statement should be added to change the display of subject weight line to 
solid black line connected by dots - representing the subject’s weight measured at that age. The SYMBOL statement 
could be:  

symbol8 interpol=join   width=1 color=black  line=1  v=diamondfilled; 

 

and updated PLOT statement should be: 

proc gplot data=all1 gout=myplot; 

Figure 4: Added reference line legends on secondary vertical axis with the same colors as the lines 

   plot (sd3neg sd2neg sd1neg sd0 sd1pos sd2pos sd3pos wt)*AGE / overlay frame 

name="sdplt" haxis=axis1 vaxis=axis2 annotate=linetext; 

 

Below is a sample output after updated PLOT statement.  

 

 

 
 

Now we add the subject exposure and safety information on the above chart. Study product start (SPSTART), study 
product stop (SPSTOP) and early termination (EARLTERM) information are represented by solid vertical lines of 
different colors at age of their occurrence. 

AE preferred term (AEPT) and CM generic name (CMTERM) are represented by dotted vertical lines of red and blue 
colors respectively at age of their occurrence. 

Serious adverse events (AEPT at AESR=’YES’) are represented by solid vertical lines with the same color as AEPT. 
In case an AE start date and CM start date occur on the same day, both shares the same line. CM terms are 
displayed in downward step style from the top of the chart area. (S)AE terms are displayed in upward step style from 
bottom of chart area. Please see Figure 1.  

The above information can be added in the chart by amending the annotate dataset as below: 

 /* Create the Annotate data set LINETEXT */ 

data linetext;*(drop=age sex sd:);    

  length function color $ 10 text $ 30 style $ 25; 

  retain xsys '2' ysys '1' hsys '1' position '6'  size 1; 

         

  set all1 end=last; 

  style = "'Albany AMT/bold'"; 

  if SPstart ne '' then do; 

    function='move';  x=age; y=0;   output; 

 function='draw';  x=age; y=100; size=.1; color='Green'; line=1; output; 

Figure 5: Added subject’s weight in the plot 

 function='label'; x=age+.5; y=97;  size=2.2; color='Green'; angle=0;    

    text=strip('Study Product Start'); output;  

  end; 

  if SPstop ne '' then do; 

 function='move';  x=age; y=0;   output; 

 function='draw';  x=age; y=100; size=.1; color='Goldenrod'; line=1; output; 

 function='label'; x=age+.5; y=97;  size=2.2; color='Goldenrod'; angle=0;  

    text=strip('Study Product Stop'); output;  

  end; 

  if earlterm ne '' then do; 

 function='move';  x=age; y=0;   output; 

 function='draw';  x=age; y=100; size=.1; color='hotpink'; line=1; output; 

 function='label'; x=age+.5; y=7;  size=2.2; color='hotpink'; angle=0;   

    text=strip('Early Termination'); output;  

  end; 

  cborder=''; 

  if aeptx ne '' then a+1; 

  if cmptx ne '' then c+1; 

 

     /*  AE and CM both occurs at same age*/ 

     if aeptx ne '' and cmptx ne '' then do; 

 if aeptx ne '' and aeser ne '' then do; 

   function='move';  x=age; y=0;   output; 

   function='draw';  x=age; y=50; size=.1; color='Red'; line=1; output; 

   function='label'; x=age+.8; y=5+(a*3);  size=2; color='Red'; angle=0;  

      text=upcase(strip(aeptx)); output;  

 end; 

 else if aeptx ne '' then do; 

   function='move';  x=age; y=0;   output; 

   function='draw';  x=age; y=50; size=.1; color='Red'; line=2; output; 

   function='label'; x=age+.8; y=5+(a*3);  size=2; color='Red'; angle=0;  

      text=upcase(strip(aeptx)); output;  

 end; 

 if cmptx ne '' then do; 

   function='move';  x=age; y=51;   output; 

   function='draw';  x=age; y=100; size=.1; color='Blue'; line=2; output; 

   function='label'; x=age+.8; y=95-(c*3);  size=2; color='Blue'; angle=0;  

      text=upcase(strip(cmptx)); output;  

 end; 

  end; 

 

  /*  AE or CM occurs at different ages*/ 

  else do; 

    if aeptx ne '' then do; 

   if aeptx ne '' and aeser ne '' then do; 

     function='move';  x=age; y=0;   output; 

     function='draw';  x=age; y=100; size=.1; color='Red'; line=1; output; 

     function='label'; x=age+.8; y=5+(a*3);  size=2; color='Red'; angle=0;  

        text=upcase(strip(aeptx)); output;  

   end; 

   else if aeptx ne '' then do; 

     function='move';  x=age; y=0;   output; 

     function='draw';  x=age; y=100; size=.1; color='Red'; line=2; output; 

     function='label'; x=age+.8; y=5+(a*3);  size=2; color='Red'; angle=0;  

        text=upcase(strip(aeptx)); output;  

   end; 

 end; 

    if cmptx ne '' then do; 

   function='move';  x=age; y=1;   output; 

   function='draw';  x=age; y=100; size=.1; color='Blue'; line=2; output; 

   function='label'; x=age+.8; y=95-(c*3);  size=2; color='Blue'; angle=0;  

      text=upcase(strip(cmptx)); output;  

 end; 

  end; 

 

  if last then do; 

    xsys='2'; ysys='2'; hsys='3'; position='6'; size=2.5; 

    x=age;angle=50; 

    y=sd3neg; text=' -3 SD';  color='brown';  output; 

    y=sd2neg; text=' -2 SD';  color='red';    output; 

    y=sd1neg; text=' -1 SD';  color='orange'; output; 

    y=sd0;    text=' Median'; color='green';  output; 

    y=sd1pos; text=' +1 SD';  color='orange'; output; 

    y=sd2pos; text=' +2 SD';  color='red';    output; 

    y=sd3pos; text=' +3 SD';  color='brown';  output; 

  end; 

run; 

 

The PROC GPLOT statement with the update annotated dataset would give the chart displayed in Figure 1. Similar 
charts can be prepared for female subjects and other anthropometric parameters.  

CONCLUSION 

The paper presented a step-by-step procedure for plotting a subject level trajectory of an anthropometric parameter 
values on the population based growth chart. The chart also displays information on potential covariates affecting the 
subject’s growth such as study product start/stop date, onset of adverse events, intake of concomitant medications. 
This subject level chart could be useful for monitoring the subject’s growth in the context of selected anthropometric 
parameter during the trial period, especially in child nutritional trials. This idea can be extended to other therapeutic 
areas and/or clinical monitoring - such as monitoring laboratory values over time with reference to normal ranges.  

